Soil Response to Earthquake Excitation—SRES

When the thickness, unit volume weight, shear modulus, and damping coefficient of each layer in a horizontally stratified ground are given and the input acceleration time history at the top of any reference layer is specified, the program SRES (**S**oil **Res**ponse to Earthquake Excitation) is a subroutine subprogram that calculates the response acceleration time history at the top of any objective layer.

SRES (**S**oil **Res**ponse to Earthquake Excitation)

【Purpose】

When an input acceleration time history at the top of the specified reference layer in a horizontally stratified ground, the acceleration response time history and its maximum value at the top of the specified objective layer are calculated using the frequency response function.

【Usage】

(1) How to connect

```
CALL SRES (L, TH, UW, G, ALPHA, BETA, ND1, LOBJ, LREF, DT, NN, DDY, ACC, ND2,
           ACMAX, IND)
```

Argument	Type	Parameter in calling program	Return Parameter
L	I	Total number of layers including bottom stratum (L .LE. 60)	Unchanged
TH	R 1-D array (ND1)	Thickness of each layer (unit : m)	Unchanged
UW	R 1-D array (ND1)	Unit volume weight of each layer (unit : tf/m^3)	Unchanged
G	R 1-D array (ND1)	Elastic shear modulus of each layer (unit : tf/m^2)	Unchanged
ALPHA	R 1-D array (ND1)	Scatter damping coefficient of each layer (unit : 1/sec)	Unchanged
BETA	R 1-D array (ND1)	Material damping coefficient of each layer (unit : nondimension)	Unchanged

ND1	I	Dimension size of TH, UW, G, ALPHA, and BETA in calling program	Unchanged
LOBJ	I	Number of objective layer	Unchanged
LREF	I	Number of reference layer	Unchanged
DT	R	Time increment of acceleration time history (unit : Sec)	Unchanged
NN	I	Total number of acceleration data (NN .LE. 8192)	Unchanged
DDY	R 1-D array (ND2)	Acceleration time history at surface of reference layer (unit : Gal)	Unchanged
ACC	R 1-D array (ND2)	No need to input here	Acceleration time history at surface of objective layer (unit : Gal)
ND2	I	Dimension size of DDY and ACC in calling program (ND2 .GE. NN)	Unchanged
ACMAX	R	No need to input here	Maximum acceleration at surface of objective layer (ACC) (unit : Gal)
IND	I	See the following Table	Unchanged

IND		Reference Layer	
		Internal	Exposed
Objective Layer	Internal	11	12
	Exposed	21	-



(2) Necessary subroutines and function subprograms

FESP FAST

(3) Remarks

i) Layer thickness of the bottom stratum *TH*(*L*) is not necessary.

ii) The ground surface is always considered the internal boundary surface.

iii) If *IND* = 21, the objective layer must be located deeper than the reference layer.

iv) The case that both the reference and objective layers are exposed boundaries is not permitted.

【Calculation Method】

The calculation is carried out in the following steps.

i) The acceleration time history $\ddot{\xi}s\ (t)$ of the input seismic motion given to the reference layer *s* is Fourier transformed by the Fast Fourier Transform program **FAST**, after adding subsequent zeros if necessary, and the harmonic vibration component $F_s(\omega)$ is obtained.

ii) The frequency response function $Zr/s(\omega)$ of the objective layer r to the reference layer s is calculated by the program **FESP** using the given thickness *H*, unit volume weight γ, shear modulus *G*, and damping factors *alpha* and *beta* of each layer.

iii) Multiply $Zr/s(\omega)$ by $F_S(\omega)$ to obtain the harmonic vibration component of the response, $F_r(\omega)$.

iv) The time history of the response acceleration $\ddot{\xi}_r(t)$ in the objective layer *r* is obtained by performing an inverse Fourier transform of $F_r(\omega)$ using the program **FAST** again.

v) Find the maximum value of the response acceleration time history $\ddot{\xi}_r(t)$.

Both the reference and the objective layers can be treated as internal or exposed boundary surfaces. The combination of the two layers can be specified by the value of the argument *IND*, but neither the reference layer nor the objective layer can be exposed boundary surface. And if the objective layer is exposed and the reference layer is internal (*IND* = 21), there are two things to note. One is that the objective layer must be located deeper than the reference layer. The other is that this program calls subroutine **FESP** to calculate the frequency response function, but **FESP** cannot handle the case where the objective layer is exposed. So, in this case, the frequency response function is calculated assuming that the objective layer is internal, and the reference layer is exposed (*IND*2 = 1 in **FESP**), and the inverse of the obtained frequency response function is computed.

This program assumes that the shear modulus and damping coefficient of each layer are constant, i.e., not strain-dependent. Therefore, this program can only be applied in the case of small seismic motions where the maximum shear strain of each layer is almost within 0.01% for all layers. The method of calculating the maximum shear strain is shown in [Example 2] of the program **FESP**.

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * *             SRES   1
C     SUROUTINE FOR SOIL RESPONSE TO EARTHQUAKE MOTION                    SRES   2
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * *             SRES   3
C                                                                         SRES   4
C                                     CODED BY Y.OHSAKI                   SRES   5
C                                                                         SRES   6
C     PURPOSE                                                             SRES   7
C       TO COMPUTE, BY MEANS OF FREQUENCY RESPONSE FUNCTION, THE RES-     SRES   8
C       PONSE ACCELERATION TIME-HISTORY AT THE TOP OF THE SPECIFIED       SRES   9
C       OBJECTIVE LAYER IN A HORIZONTALLY LAYERED SOIL DEPOSIT, WHEN      SRES  10
C       AN INPUT ACCELERATION TIME-HISTORY IS GIVEN AT THE TOP OF THE     SRES  11
C       SPECIFIED REFERENCE LAYER.                                        SRES  12
C                                                                         SRES  13
C     USAGE                                                               SRES  14
C       CALL SRES(L,TH,UW,G,ALPHA,BETA,ND1,LOBJ,LREF,DT,NN,DDY,ACC,       SRES  15
C                ND2,ACMAX,IND)                                           SRES  16
C                                                                         SRES  17
C     DESCRIPTION OF ARGUMENTS                                            SRES  18
C       L           - TOTAL NUMBER OF LAYERS, INCLUDING THE BASE LAYER    SRES  19
C                     L.LE.60                                             SRES  20
C       TH(ND1)     - THICKNESS IN METERS                                 SRES  21
C       UW(ND1)     - UNIT WEIGHT IN TONS/CU.METER                        SRES  22
C       G(ND1)      - SHEAR MODULUS IN TONS/SQ.METER                      SRES  23
C       ALPHA(ND1)  - SCATTER DAMPING COEFFICIENT IN 1/SEC                SRES  24
```

```
C       BETA(ND1)  - MATERIAL DAMPING COEFFICIENT IN DECIMAL FRACTION   SRES
C       ND1        - DIMENSION OF TH,UW,G,ALPHA BETA IN CALLING PROGRAM SRES
C       LOBJ       - NUMBER OF OBJECTIVE LAYER                          SRES
C       LREF       - NUMBER OF REFERENCE LAYER                          SRES
C       DT         - TIME INCREMENT IN SEC                              SRES
C       NN         - TOTAL NUMBER OF DATA IN THE TIME-HISTORIES         SRES
C                    NN.LE.8192                                         SRES
C       DDY(ND2)   - REFERENCE ACCELERATION TIME-HISTORY IN GALS        SRES
C       ACC(ND2)   - OBJECTIVE ACCELERATION TIME-HISTORY IN GALS        SRES
C       ND2        - DIMENSION OF DDY,ACC IN CALLING PROGRAM            SRES
C       ACMAX      - MAX. ACCELERATION OF ACC IN GALS                   SRES
C       IND        - SEE THE FOLLOWING TABLE                            SRES
C                  ---------------------------------------------------  SRES
C                  I                            I  REFERENCE LAYER   I  SRES
C                  I            IND             I--------------------I  SRES
C                  I                            I INTERNAL I EXPOSED I  SRES
C                  I----------------------------I----------I---------I  SRES
C                  I                  I INTERNAL I    11    I    12   I  SRES
C                  I OBJECTIVE LAYER I----------I----------I---------I  SRES
C                  I                  I EXPOSED  I    21    I    NO   I  SRES
C                  ---------------------------------------------------  SRES
C                                                                       SRES
C     REMARKS                                                           SRES
C       (1)  THICKNESS OF BASE LAYER TH(L)  IS NOT REQUIRED             SRES
C       (2)  THE GROUND SURFACE IS ALWAYS REGARDED AS INTERNAL          SRES
C       (3)  WHEN IND=21,THE OBJECTIVE LAYER MUST BE LOCATED DEEPER     SRES
C            THAN THE REFERENCE LAYER                                   SRES
C       (4)  THE CASE THAT BOTH REFERENCE AND OBJECTIVE LAYERS ARE      SRES
C            EXPOSED IS NOT PERMITTED                                   SRES
C                                                                       SRES
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED                     SRES
C       FESP FAST                                                       SRES
C                                                                       SRES
      SUBROUTINE SRES(L,TH,UW,G,ALPHA,BETA,ND1,LOBJ,LREF,DT,NN,DDY,     SRES
     *           ACC,ND2,ACMAX,IND)                                     SRES
C                                                                       SRES
      DIMENSION TH(ND1),UW(ND1),G(ND1),ALPHA(ND1),BETA(ND1),DDY(ND2),   SRES
     *          ACC(ND2)                                                SRES
      COMPLEX   H(4097),C(8192)                                         SRES
C                                                                       SRES
C     FOURIER TRANSFORM OF REFERENCE TIME-HISTORY                       SRES
C                                                                       SRES
      NT=2                                                              SRES
  110 IF(NT.GE.NN) GO TO 120                                            SRES
      NT=NT*2                                                           SRES
      GO TO 110                                                         SRES
  120 NFOLD=NT/2+1                                                      SRES
      DO 130 M=1,NN                                                     SRES
      C(M)=CMPLX(DDY(M)/REAL(NT),0.)                                    SRES
  130 CONTINUE                                                          SRES
      IF(NN.EQ.NT) GO TO 150                                            SRES
      DO 140 M=NN+1,NT                                                  SRES
      C(M)=(0.,0.)                                                      SRES
```

```
  140 CONTINUE
  150 CALL FAST(NT,C,8192,-1)
C
C      FREQUENCY RESPONSE FUNCTION
C
      DF=1./REAL(NT)/DT
      IF(IND.EQ.21) GO TO 160
      CALL FESP(L,TH,UW,G,ALPHA,BETA,ND1,LOBJ,LREF,DF,NFOLD,H,4097,0,
     *         IND-11)
      GO TO 180
  160 CALL FESP(L,TH,UW,G,ALPHA,BETA,ND1,LREF,LOBJ,DF,NFOLD,H,4097,0,1)
      DO 170 K=1,NFOLD
      H(K)=(1.,0.)/H(K)
  170 CONTINUE
C
C      OBJECTIVE TIME-HISTORY
C
  180 C(1)=C(1)*H(1)
      DO 190 K=2,NFOLD-1
      C(K)=C(K)*H(K)
      C(NT-K+2)=CONJG(C(K))
  190 CONTINUE
      C(NFOLD)=C(NFOLD)*H(NFOLD)
      CALL FAST(NT,C,8192,+1)
      ACMAX=0.
      DO 200 M=1,NN
      ACC(M)=REAL(C(M))
      ACMAX=AMAX1(ACMAX,ABS(ACC(M)))
  200 CONTINUE
      RETURN
      END
```

【Example1】

For the ground shown in the table below, read the acceleration time history observed at the top of the base stratum at a depth of 10.90 m from the file EQ.03. Give it as input and compute the response acceleration time history at the ground surface.

Layer No.	Upper depth (m)	Layer thickness (m)	Unit volume weight (tf/m^3)	Elastic shear modulus (tf/m^2)
1	0.00	3.80	1.50	1200
2	3.80	3.20	1.67	2900
3	7.00	3.90	1.85	5700
4	10.90	-	1.95	50000

```
C
        DIMENSION TH(4),UW(4),G(4),ALPHA(4),BETA(4),DDY(1000),
      *           ACC(1000)
        DATA      L/4/,TH/3.8,3.2,3.9,0.0/,UW/1.50,1.67,1.85,1.95/,
      *           G/1200.,2900.,5700.,50000/,ALPHA/4*2.0/,BETA/4*0.02/,
      *           LOBJ/1/,LREF/4/,IND/11/
C
        READ(5,501) DT,NN,(DDY(M),M=1,NN)
        CALL SRES(L,TH,UW,G,ALPHA,BETA,4,LOBJ,LREF,DT,NN,DDY,ACC,1000,
      *           ACMAX,IND)
        STOP
C
  501 FORMAT(T51,F10.0,I10/(8F10.0) )
        END
```

Output: Calculated results are stored in the array *ACC* and plotted as the blue line in the following figure. The acceleration time history in the file EQ.02 (recorded at the ground surface at the same time) is also shown in this figure in gray line for reference.



Note: The maximum response acceleration is different from that in the Japanese version manual. This is because the value of attenuation *ALPA* was set to 2.0 in the English version, the same as in Dr. Ohsaki's book, while it was 1.8 in the Japanese version. This is also true for the following example.

【Example2】

Using the same ground model as in Example 1, read the acceleration time history on the ground surface from file EQ.02, and remove the surface soil up to a depth of 10.90 m to expose the upper plane of the base stratum and calculate the acceleration time history at this plane.

```
C
      DIMENSION TH(4),UW(4),G(4),ALPHA(4),BETA(4),DDY(1000),
     *          ACC(1000)
      DATA      L/4/,TH/3.8,3.2,3.9,0.0/,UW/1.50,1.67,1.85,1.95/,
     *          G/1200.,2900.,5700.,50000/,ALPHA/4*2.0/,BETA/4*0.02/,
     *          LOBJ/4/,LREF/1/,IND/21/
C
      READ(5,501) DT,NN,DDYMAX,(DDY(M),M=1,NN)
      WRITE(6,601) DDYMAX
      CALL SRES(L,TH,UW,G,ALPHA,BETA,4,LOBJ,LREF,DT,NN,DDY,ACC,1000,
     *          ACMAX,IND)
      WRITE(6,602) ACMAX
      STOP
C
  501 FORMAT(T51,F10.0,I10,F10.0/(8F10.0) )
  601 FORMAT('MAX ACCELERATION'/T3,'GROUND SURFACE',F10.2,TR1,'(GAL)')
  602 FORMAT(T3,'EXPOSED ROCK  ',F10.2,TR1,'(GAL)')
      END
```

Output: Calculated results are stored in the array *ACC* and the maximum acceleration is 8.94 Gal as shown below.

```
MAX ACCELERATION
  GROUND SURFACE     19.70 (GAL)
  EXPOSED ROCK        8.94 (GAL)
```